Cartan equivalence under fibre preserving transformations for second order ODE : $u_{x,x} = F(x, u, u_x)$

First load exterior and create the appropriate jet bundle

```
> with(Exterior):
```

Exterior calculus package, version 1.12 (30 Oct 2009).

 (1)

```
> jetbundle([x],[u],2);
```

Next we create the contact forms and the exterior differential system for this equation

```
> depend([x,u,u[x]],F):
> Contact();
```

$$-u_x\,dx + du,\ -u_{x,x}\,dx + du_x \tag{2}$$

```
> EDS:=eval({%},u[x,x]=F);
```

$$EDS := \{-u_x\,dx + du,\ -F\,dx + du_x\} \tag{3}$$

We can differentiate this system and then check that it is closed

```
> d(%);
```

$$\left\{\left(\frac{\partial}{\partial u_x} F\right) dx \wedge (du_x) + \left(\frac{\partial}{\partial u} F\right) dx \wedge du,\ dx \wedge (du_x)\right\} \tag{4}$$

```
> ideal(EDS);
```

$$\{du = u_x\,dx,\ du_x = F\,dx\} \tag{5}$$

```
> Mod(%%,%);
```

$$\{0\} \tag{6}$$

Now for the coframe

```
> omega:=coframe(Diff(u,x,x)=F);
```

$$\omega := \begin{bmatrix} -u_x\,dx + du \\ -F\,dx + du_x \\ dx \end{bmatrix} \tag{7}$$

and it dual.

```
> dual_omega:=dual(omega);
```

$$dual\_omega := \begin{bmatrix} \partial_u \\ \partial_{u_x} \\ \partial_x + u_x\,\partial_u + F\,\partial_{u_x} \end{bmatrix} \tag{8}$$

```
> dual_omega.Transpose(omega);
```

(9)

$$\begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix} \quad \textbf{(9)}$$

Just checking!

```
> d(omega);
```

$$\begin{bmatrix} dx \wedge (du_x) \\ \left(\frac{\partial}{\partial u_x} F\right) dx \wedge (du_x) + \left(\frac{\partial}{\partial u} F\right) dx \wedge du \\ 0 \end{bmatrix} \quad \textbf{(10)}$$

We now need to construct the lifted coframe. First we need to add the group parameters to the exterior algebra.

```
> exterior(a[1],a[2],a[3],a[4]);
> EXTERIOR:-var;
```

$$[x, u, u_x, u_{x,x}, a_1, a_2, a_3, a_4] \quad \textbf{(11)}$$

This shows the current variables in the exterior algebra. Fibre preserving transformations are given by

```
> Fibre:=<<a[1]|0|0>,<a[2]|a[3]|0>,<0|0|a[4]>>;
```

$$Fibre := \begin{bmatrix} a_1 & 0 & 0 \\ a_2 & a_3 & 0 \\ 0 & 0 & a_4 \end{bmatrix} \quad \textbf{(12)}$$

with a basis of Maurer-Cartan forms

```
> MC:=Maurer_Cartan(Fibre);
```

$$MC := \begin{bmatrix} \frac{da_1}{a_1} \\ \frac{da_2}{a_1} - \frac{a_2\, da_3}{a_1\, a_3} \\ \frac{da_3}{a_3} \\ \frac{da_4}{a_4} \end{bmatrix} \quad \textbf{(13)}$$

or in "matrix" form

```
> MC:=Maurer_Cartan(Fibre,matrixonly);
```

(14)

$$MC := \begin{bmatrix} \frac{da_1}{a_1} & 0 & 0 \\ \frac{da_2}{a_1} - \frac{a_2\, da_3}{a_1\, a_3} & \frac{da_3}{a_3} & 0 \\ 0 & 0 & \frac{da_4}{a_4} \end{bmatrix} \qquad \textbf{(14)}$$

The lifted coframe and its dual are given by

```
> theta:=Fibre.omega;
```

$$\theta := \begin{bmatrix} -a_1\, u_x\, dx + a_1\, du \\ \left(-a_2\, u_x - a_3\, F\right)\, dx + a_2\, du + a_3\, du_x \\ a_4\, dx \end{bmatrix} \qquad \textbf{(15)}$$

```
> dual_theta:=dual(theta);
```

$$dual\_theta := \begin{bmatrix} \frac{\partial_u}{a_1} - \frac{a_2\, \partial_{u_x}}{a_1\, a_3} \\ \frac{\partial_{u_x}}{a_3} \\ \frac{\partial_x}{a_4} + \frac{u_x\, \partial_u}{a_4} + \frac{F\, \partial_{u_x}}{a_4} \end{bmatrix} \qquad \textbf{(16)}$$

The essential torsion of this frame is given by

```
> ET,Phi,Z,tableau:=torsion(omega,group=Fibre,essential);
```

$$ET, \Phi, Z, tableau := \begin{bmatrix} \begin{bmatrix} 0 & 0 & 0 \\ 0 & 0 & -\frac{a_1}{a_4\, a_3} \\ 0 & 0 & 0 \end{bmatrix} \\ \begin{bmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{bmatrix} \\ \begin{bmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{bmatrix} \end{bmatrix}, \qquad \textbf{(17)}$$

$$\left[\begin{array}{c} -\frac{a_2\,dx}{a_3}+\frac{da_1}{a_1} \\ \frac{\left(a_3^2\left(\frac{\partial}{\partial u}F\right)-a_2^2-a_2\,a_3\left(\frac{\partial}{\partial u_x}F\right)\right)dx}{a_1\,a_3}+\frac{da_2}{a_1}-\frac{a_2\,da_3}{a_1\,a_3} \\ \frac{\left(a_2+a_3\left(\frac{\partial}{\partial u_x}F\right)\right)dx}{a_3}+\frac{da_3}{a_3} \\ \frac{da_4}{a_4} \end{array}\right], \left[\begin{array}{ccc} \chi_{1,1} & 0 & 0 \\ \chi_{2,1} & \chi_{3,1} & 0 \\ \chi_{3,1} & \chi_{3,2} & 0 \\ 0 & 0 & \chi_{4,3} \end{array}\right],$$

$$\left[\begin{array}{cccc} v_1 & 0 & 0 & 0 \\ 0 & v_1 & v_2 & 0 \\ 0 & 0 & 0 & v_3 \end{array}\right]$$

ET is the essential torsion; Φ are the modified Maurer-Cartan forms that remove the non-essential torsion; Z is the freedom in this coframe and tableau is the coefficient matrix for the Maurer-Cartan forms in the structure equations (used to compute the reduced Cartan characters). The absorbed form of the coframe satisfies

```
> AbsorbedForm(theta,Phi,ET,tableau);
```

$$d\theta=\left[\begin{array}{c} (\Phi_1)\wedge(\theta_1)-\frac{a_1\,(\theta_2)\wedge(\theta_3)}{a_4\,a_3} \\ (\Phi_2)\wedge(\theta_1)+(\Phi_3)\wedge(\theta_2) \\ (\Phi_4)\wedge(\theta_3) \end{array}\right] \qquad \textbf{(18)}$$

or with the freedom in the absorbed frame

```
> AbsorbedForm(theta,Phi,ET,tableau,freedom=Z);
```

$$d\theta=\left[\begin{array}{c} (\Phi_1)\wedge(\theta_1)-\frac{a_1\,(\theta_2)\wedge(\theta_3)}{a_4\,a_3} \\ (\Phi_2)\wedge(\theta_1)+(\Phi_3)\wedge(\theta_2) \\ (\Phi_4)\wedge(\theta_3) \end{array}\right], \Phi\rightarrow\left(\left[\begin{array}{c} \Phi_1+\chi_{1,1}\,\theta_1 \\ \Phi_2+\chi_{2,1}\,\theta_1+\chi_{3,1}\,\theta_2 \\ \Phi_3+\chi_{3,1}\,\theta_1+\chi_{3,2}\,\theta_2 \\ \Phi_4+\chi_{4,3}\,\theta_3 \end{array}\right]\right) \qquad \textbf{(19)}$$

The only (non-constant) essential torsion is

```
> nc_torsion(ET);
```

$$\left\{-\frac{a_1}{a_4\,a_3}\right\} \qquad \textbf{(20)}$$

We normalize this essential torsion to -1 by

```
> exterior(a[3]=a[1]/a[4]);
> Phi[3];
```

$$\frac{a_4 \left(a_2 + \frac{a_1 \left(\frac{\partial}{\partial u_x} F \right)}{a_4} \right) dx}{a_1} + \frac{a_4 \left(\frac{da_1}{a_4} - \frac{a_1\, da_4}{a_4^2} \right)}{a_1} \tag{21}$$

```
> Fibre;
```

$$\begin{bmatrix} a_1 & 0 & 0 \\ a_2 & \frac{a_1}{a_4} & 0 \\ 0 & 0 & a_4 \end{bmatrix} \tag{22}$$

```
> Maurer_Cartan(Fibre);
```

$$\begin{bmatrix} \frac{da_1}{a_1} \\ -\frac{a_2\, da_1}{a_1^2} + \frac{da_2}{a_1} + \frac{a_2\, da_4}{a_4\, a_1} \\ \frac{da_4}{a_4} \end{bmatrix} \tag{23}$$

Recompute the torsion.

```
> ET,Phi,Z,tableau:=torsion(omega,group=Fibre,essential);
```

$$ET, \Phi, Z, tableau := \begin{bmatrix} \begin{bmatrix} 0 & 0 & 0 \\ 0 & 0 & -1 \\ 0 & 0 & 0 \end{bmatrix} \\ \begin{bmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{bmatrix} \\ \begin{bmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{bmatrix} \end{bmatrix}, \tag{24}$$

$$\left[\begin{array}{c} -\frac{a_2 a_4 dx}{a_1} + \frac{da_1}{a_1} \\ \frac{\left(\left(\frac{\partial}{\partial u} F\right) a_1^2 - a_2^2 a_4^2 - a_2 a_4 a_1 \left(\frac{\partial}{\partial u_x} F\right)\right) dx}{a_4 a_1^2} - \frac{a_2 da_1}{a_1^2} + \frac{da_2}{a_1} + \frac{a_2 da_4}{a_4 a_1} \\ -\frac{\left(2 a_2 a_4 + a_1 \left(\frac{\partial}{\partial u_x} F\right)\right) dx}{a_1} + \frac{da_4}{a_4} \end{array}\right],$$

$$\left[\begin{array}{ccc} \chi_{2,2} & 0 & 0 \\ \chi_{2,1} & \chi_{2,2} & 0 \\ 0 & 0 & 0 \end{array}\right], \left[\begin{array}{ccc} v_1 & 0 & 0 \\ v_2 & v_1 & -v_2 \\ 0 & 0 & v_3 \end{array}\right]$$

```
> AbsorbedForm(theta,Phi,ET,tableau,freedom=Z);
```

$$d\theta = \left[\begin{array}{c} (\Phi_1) \wedge (\theta_1) - (\theta_2) \wedge (\theta_3) \\ (\Phi_1) \wedge (\theta_2) + (\Phi_2) \wedge (\theta_1) - (\Phi_3) \wedge (\theta_2) \\ (\Phi_3) \wedge (\theta_3) \end{array}\right], \Phi \to \left(\left[\begin{array}{c} \Phi_1 + \chi_{2,2}\theta_1 \\ \Phi_2 + \chi_{2,1}\theta_1 + \chi_{2,2}\theta_2 \\ \Phi_3 \end{array}\right]\right) \quad \mathbf{(25)}$$

```
> CartanCharacters(tableau);
```

$$[3, 0, 0] \quad \mathbf{(26)}$$

```
> CartanTest(tableau,Z);
```

System is NOT involutive. **(27)**

The system fails the Cartan test. Therefore we must prolong. Add the freedeom in the coframe $\chi_{2,1}$ and $\chi_{2,2}$ to the exterior algebra.

```
> exterior(op(indets(Z)));
> EXTERIOR:-var;
```

$$[x, u, u_x, u_{x,x}, a_1, a_2, a_4, \chi_{2,1}, \chi_{2,2}] \quad \mathbf{(28)}$$

Prolonged group action is given by

```
> ProlongedGroup:=ProlongedAction(Z);
```

(29)

$$ProlongedGroup := \begin{bmatrix} 1 & 0 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 & 0 \\ \chi_{2,2} & 0 & 0 & 1 & 0 & 0 \\ \chi_{2,1} & \chi_{2,2} & 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 & 0 & 1 \end{bmatrix} \quad \textbf{(29)}$$

Prolonged coframe

```
> Theta:=Vector([theta,Phi]);
```

$$\Theta := \begin{bmatrix} -a_1 u_x dx + a_1 du \\ \left(-a_2 u_x - \frac{F a_1}{a_4}\right) dx + a_2 du + \frac{a_1 du_x}{a_4} \\ a_4 dx \\ -\frac{a_2 a_4 dx}{a_1} + \frac{da_1}{a_1} \\ \frac{\left(\left(\frac{\partial}{\partial u} F\right) a_1^2 - a_2^2 a_4^2 - a_2 a_4 a_1 \left(\frac{\partial}{\partial u_x} F\right)\right) dx}{a_4 a_1^2} - \frac{a_2 da_1}{a_1^2} + \frac{da_2}{a_1} + \frac{a_2 da_4}{a_4 a_1} \\ -\frac{\left(2 a_2 a_4 + a_1 \left(\frac{\partial}{\partial u_x} F\right)\right) dx}{a_1} + \frac{da_4}{a_4} \end{bmatrix} \quad \textbf{(30)}$$

The torsion is

```
> T1,pi,Z1,tableau1:=torsion(Theta,group=ProlongedGroup,essential,
  absorption=mu):
> AbsorbedForm(Theta,pi,T1,tableau1,freedom=Z1);
```

$$d\Theta = \Big[\Big[-(\Theta_1) \wedge (\Theta_4) - (\Theta_2) \wedge (\Theta_3)\Big], \quad \textbf{(31)}$$

$$\Big[-(\Theta_1) \wedge (\Theta_5) - (\Theta_2) \wedge (\Theta_4) + (\Theta_2) \wedge (\Theta_6)\Big],$$

$$\Big[-(\Theta_3) \wedge (\Theta_6)\Big],$$

$$\left[(\pi_2) \wedge (\Theta_1) + (\Theta_3) \wedge (\Theta_5)\right],$$

$$\left[(\pi_1) \wedge (\Theta_1) + (\pi_2) \wedge (\Theta_2) + \frac{\left(-2\,\chi_{2,1}\,a_1^2\,a_4 + \left(\frac{\partial^2}{\partial u_x\,\partial u} F\right) a_1 - a_2\,a_4 \left(\frac{\partial^2}{\partial u_x^2} F\right)\right) (\Theta_2) \wedge (\Theta_3)}{a_4\,a_1^2} + (\Theta_5) \wedge (\Theta_6)\right],$$

$$\left[-\frac{\left(-2\,\chi_{2,1}\,a_1^2\,a_4 + \left(\frac{\partial^2}{\partial u_x\,\partial u} F\right) a_1 - a_2\,a_4 \left(\frac{\partial^2}{\partial u_x^2} F\right)\right) (\Theta_1) \wedge (\Theta_3)}{a_4\,a_1^2} - \frac{\left(-2\,\chi_{2,2}\,a_1 + \frac{\partial^2}{\partial u_x^2} F\right) (\Theta_2) \wedge (\Theta_3)}{a_1} + 2\,(\Theta_3) \wedge (\Theta_5)\right]\Bigg],$$

$$\pi \to \left(\begin{bmatrix} \pi_1 + \mu_{1,1}\,\Theta_1 + \mu_{2,1}\,\Theta_2 \\ \pi_2 + \mu_{2,1}\,\Theta_1 \end{bmatrix}\right)$$

```
> nc_torsion(T1);
```

$$\left\{\frac{-2\,\chi_{2,1}\,a_1^2\,a_4 + \left(\frac{\partial^2}{\partial u_x\,\partial u} F\right) a_1 - a_2\,a_4 \left(\frac{\partial^2}{\partial u_x^2} F\right)}{a_4\,a_1^2}, -\frac{-2\,\chi_{2,2}\,a_1 + \frac{\partial^2}{\partial u_x^2} F}{a_1}, -\frac{-2\,\chi_{2,1}\,a_1^2\,a_4 + \left(\frac{\partial^2}{\partial u_x\,\partial u} F\right) a_1 - a_2\,a_4 \left(\frac{\partial^2}{\partial u_x^2} F\right)}{a_4\,a_1^2}\right\} \quad \textbf{(32)}$$

The non-constant torsion components may be set to zero with the normalizations

```
> isolate(%[1],chi[2,1]);
```

$$\chi_{2,1} = -\frac{1}{2}\,\frac{-\left(\frac{\partial^2}{\partial u_x\,\partial u} F\right) a_1 + a_2\,a_4 \left(\frac{\partial^2}{\partial u_x^2} F\right)}{a_1^2\,a_4} \quad \textbf{(33)}$$

```
> isolate(%%[2],chi[2,2]);
```

$$\chi_{2,2} = \frac{1}{2}\,\frac{\frac{\partial^2}{\partial u_x^2} F}{a_1} \quad \textbf{(34)}$$

```
> exterior(%,%%);
> Transpose(T1);
```

$$\left[\begin{bmatrix} 0&0&0&-1&0&0\\ 0&0&-1&0&0&0\\ 0&0&0&0&0&0\\ 0&0&0&0&0&0\\ 0&0&0&0&0&0\\ 0&0&0&0&0&0 \end{bmatrix}, \begin{bmatrix} 0&0&0&0&-1&0\\ 0&0&0&-1&0&1\\ 0&0&0&0&0&0\\ 0&0&0&0&0&0\\ 0&0&0&0&0&0\\ 0&0&0&0&0&0 \end{bmatrix}, \begin{bmatrix} 0&0&0&0&0&0\\ 0&0&0&0&0&0\\ 0&0&0&0&0&-1\\ 0&0&0&0&0&0\\ 0&0&0&0&0&0\\ 0&0&0&0&0&0 \end{bmatrix}, \begin{bmatrix} 0&0&0&0&0&0\\ 0&0&0&0&0&0\\ 0&0&0&0&1&0\\ 0&0&0&0&0&0\\ 0&0&0&0&0&0\\ 0&0&0&0&0&0 \end{bmatrix}, \begin{bmatrix} 0&0&0&0&0&0\\ 0&0&0&0&0&0\\ 0&0&0&0&0&0\\ 0&0&0&0&0&0\\ 0&0&0&0&0&1\\ 0&0&0&0&0&0 \end{bmatrix}, \begin{bmatrix} 0&0&0&0&0&0\\ 0&0&0&0&0&0\\ 0&0&0&0&2&0\\ 0&0&0&0&0&0\\ 0&0&0&0&0&0\\ 0&0&0&0&0&0 \end{bmatrix}\right] \quad \textbf{(35)}$$

```
> CartanCharacters(tableau1);
```

$$[2, 0, 0, 0, 0, 0] \quad \textbf{(36)}$$

```
> CartanTest(tableau1,Z1);
```

System is involutive. **(37)**

```
> pi:=map(expand,pi);
```

$$\pi := \Bigg[\Bigg[\Bigg(-\frac{\left(\frac{\partial^2}{\partial u^2} F\right) a_1^2 - 2\, a_2\, a_4\, a_1 \left(\frac{\partial^2}{\partial u_x\, \partial u} F\right) + a_2^2\, a_4^2 \left(\frac{\partial^2}{\partial u_x^2} F\right)}{a_1^3\, a_4} \quad \textbf{(38)}$$

$$- \frac{1}{2}\, \frac{\left(-\left(\frac{\partial^2}{\partial u_x\, \partial u} F\right) a_1 + a_2\, a_4 \left(\frac{\partial^2}{\partial u_x^2} F\right)\right)\left(-\frac{1}{2}\left(\frac{\partial^2}{\partial u_x^2} F\right) u_x - \frac{a_2\, a_4}{a_1}\right)}{a_1^2\, a_4}$$

$$+ \frac{1}{2}\, \frac{1}{a_1} \left(\left(\frac{\partial^2}{\partial u_x^2} F\right) \left(\frac{1}{2}\, \frac{\left(-\left(\frac{\partial^2}{\partial u_x\, \partial u} F\right) a_1 + a_2\, a_4 \left(\frac{\partial^2}{\partial u_x^2} F\right)\right) u_x}{a_1\, a_4} \right.\right.$$

$$+\frac{1}{2}\,\frac{\left(\frac{\partial^2}{\partial u_x^2}F\right)\left(-a_2\,u_x-\frac{F\,a_1}{a_4}\right)}{a_1}+\frac{\left(\frac{\partial}{\partial u}F\right)a_1^2-a_2^2\,a_4^2-a_2\,a_4\,a_1\left(\frac{\partial}{\partial u_x}F\right)}{a_4\,a_1^2}\Bigg)\Bigg)$$

$$+\frac{1}{2}\,\frac{\left(-\left(\frac{\partial^2}{\partial u_x\,\partial u}F\right)a_1+a_2\,a_4\left(\frac{\partial^2}{\partial u_x^2}F\right)\right)\left(2\,a_2\,a_4+a_1\left(\frac{\partial}{\partial u_x}F\right)\right)}{a_1^3\,a_4}$$

$$-\frac{1}{2}\,\frac{-\left(\frac{\partial^3}{\partial u_x\,\partial x\,\partial u}F\right)a_1+a_2\,a_4\left(\frac{\partial^3}{\partial u_x^2\,\partial x}F\right)}{a_1^2\,a_4}\Bigg)\,dx$$

$$+\left(\frac{1}{2}\,\frac{-\left(\frac{\partial^2}{\partial u_x\,\partial u}F\right)a_1+a_2\,a_4\left(\frac{\partial^2}{\partial u_x^2}F\right)}{a_1^3\,a_4}-\frac{1}{2}\,\frac{a_2\left(\frac{\partial^2}{\partial u_x^2}F\right)}{a_1^3}+\frac{1}{2}\,\frac{\frac{\partial^2}{\partial u_x\,\partial u}F}{a_1^2\,a_4}\right)da_1$$

$$+\Bigg(-\frac{1}{4}\,\frac{\left(\frac{\partial^2}{\partial u_x^2}F\right)\left(-\left(\frac{\partial^2}{\partial u_x\,\partial u}F\right)a_1+a_2\,a_4\left(\frac{\partial^2}{\partial u_x^2}F\right)\right)}{a_1^2\,a_4}$$

$$+\frac{1}{2}\,\frac{\left(-\frac{1}{2}\,\frac{-\left(\frac{\partial^2}{\partial u_x\,\partial u}F\right)a_1+a_2\,a_4\left(\frac{\partial^2}{\partial u_x^2}F\right)}{a_1\,a_4}+\frac{1}{2}\,\frac{\left(\frac{\partial^2}{\partial u_x^2}F\right)a_2}{a_1}\right)\left(\frac{\partial^2}{\partial u_x^2}F\right)}{a_1}$$

$$-\frac{1}{2}\,\frac{-\left(\frac{\partial^3}{\partial u_x\,\partial u^2}F\right)a_1+a_2\,a_4\left(\frac{\partial^3}{\partial u_x^2\,\partial u}F\right)}{a_1^2\,a_4}\Bigg)\,du+\left(\frac{1}{4}\,\frac{\left(\frac{\partial^2}{\partial u_x^2}F\right)^2}{a_1\,a_4}\right.$$

$$\left.-\frac{1}{2}\,\frac{-\left(\frac{\partial^3}{\partial u_x^2\,\partial u}F\right)a_1+a_2\,a_4\left(\frac{\partial^3}{\partial u_x^3}F\right)}{a_1^2\,a_4}\right)du_x\Bigg],$$

$$\left[\left(\frac{1}{2}\,\frac{-\left(\frac{\partial^2}{\partial u_x\,\partial u}F\right)a_1+a_2\,a_4\left(\frac{\partial^2}{\partial u_x^{\,2}}F\right)}{a_1^2}+\frac{1}{2}\,\frac{\left(\frac{\partial^2}{\partial u_x^{\,2}}F\right)\left(-\frac{1}{2}\left(\frac{\partial^2}{\partial u_x^{\,2}}F\right)u_x-\frac{a_2\,a_4}{a_1}\right)}{a_1}+\frac{1}{2}\,\frac{\frac{\partial^3}{\partial u_x^{\,2}\,\partial x}F}{a_1}\right)dx+\left(\frac{1}{2}\,\frac{\frac{\partial^3}{\partial u_x^{\,2}\,\partial u}F}{a_1}+\frac{1}{4}\,\frac{\left(\frac{\partial^2}{\partial u_x^{\,2}}F\right)^2}{a_1}\right)du+\frac{1}{2}\,\frac{\left(\frac{\partial^3}{\partial u_x^{\,3}}F\right)du_x}{a_1}\right]\right]$$

Invariant structure functions:

```
> J:=InvariantStructure(Theta,pi,tableau1);
```

$$J:=\left[-\frac{1}{2}\,\frac{a_4\left(\frac{\partial^3}{\partial u_x^{\,3}}F\right)}{a_1^2},\; -\frac{1}{2}\,\frac{-\left(\frac{\partial^2}{\partial u_x\,\partial u}F\right)+\frac{\partial^3}{\partial u_x^{\,2}\,\partial x}F+\left(\frac{\partial^3}{\partial u_x^{\,3}}F\right)F+u_x\left(\frac{\partial^3}{\partial u_x^{\,2}\,\partial u}F\right)}{a_1\,a_4},\; \frac{1}{2}\,\frac{1}{a_1^2\,a_4^2}\left(u_x\,a_2\,a_4\left(\frac{\partial^3}{\partial u_x^{\,2}\,\partial u}F\right)+2\left(\frac{\partial^2}{\partial u^2}F\right)a_1-a_2\,a_4\left(\frac{\partial^2}{\partial u_x\,\partial u}F\right)-\left(\frac{\partial^2}{\partial u_x^{\,2}}F\right)\left(\frac{\partial}{\partial u}F\right)a_1+\left(\frac{\partial^2}{\partial u_x\,\partial u}F\right)a_1\left(\frac{\partial}{\partial u_x}F\right)-\left(\frac{\partial^3}{\partial u_x\,\partial x\,\partial u}F\right)a_1+a_2\,a_4\left(\frac{\partial^3}{\partial u_x^{\,2}\,\partial x}F\right)-u_x\left(\frac{\partial^3}{\partial u_x\,\partial u^2}F\right)a_1+a_2\,a_4\left(\frac{\partial^3}{\partial u_x^{\,3}}F\right)F-\left(\frac{\partial^3}{\partial u_x^{\,2}\,\partial u}F\right)a_1\,F\right)\right] \tag{39}$$

Normalization (for the generic case).

```
> solve({J[1]=1,J[2]=1,J[3]=0},{a[1],a[2],a[4]});
```

$$\left\{a_1=RootOf\left(-\left(\frac{\partial^3}{\partial u_x^{\,3}}F\right)u_x\left(\frac{\partial^3}{\partial u_x^{\,2}\,\partial u}F\right)+\left(\frac{\partial^3}{\partial u_x^{\,3}}F\right)\left(\frac{\partial^2}{\partial u_x\,\partial u}F\right)-\left(\frac{\partial^3}{\partial u_x^{\,3}}F\right)\left(\frac{\partial^3}{\partial u_x^{\,2}\,\partial x}F\right)-\left(\frac{\partial^3}{\partial u_x^{\,3}}F\right)^2F+4\,\_Z^3\right),\; a_2=-\left(2\left(-2\left(\frac{\partial^2}{\partial u^2}F\right)\right.\right.\right. \tag{40}$$

$$+\left(\frac{\partial^2}{\partial u_x^2}F\right)\left(\frac{\partial}{\partial u}F\right)-\left(\frac{\partial^2}{\partial u_x\,\partial u}F\right)\left(\frac{\partial}{\partial u_x}F\right)+\frac{\partial^3}{\partial u_x\,\partial x\,\partial u}F+u_x\left(\frac{\partial^3}{\partial u_x\,\partial u^2}F\right)$$

$$+\left(\frac{\partial^3}{\partial u_x^2\,\partial u}F\right)F\Bigg)\,RootOf\Bigg(-\left(\frac{\partial^3}{\partial u_x^3}F\right)u_x\left(\frac{\partial^3}{\partial u_x^2\,\partial u}F\right)+\left(\frac{\partial^3}{\partial u_x^3}F\right)\left(\frac{\partial^2}{\partial u_x\,\partial u}F\right)$$

$$-\left(\frac{\partial^3}{\partial u_x^3}F\right)\left(\frac{\partial^3}{\partial u_x^2\,\partial x}F\right)-\left(\frac{\partial^3}{\partial u_x^3}F\right)^2F+4\,\_Z^3\Bigg)^2\Bigg)\Bigg/\Bigg(u_x^2\left(\frac{\partial^3}{\partial u_x^2\,\partial u}F\right)^2$$

$$-2\,u_x\left(\frac{\partial^3}{\partial u_x^2\,\partial u}F\right)\left(\frac{\partial^2}{\partial u_x\,\partial u}F\right)+2\,u_x\left(\frac{\partial^3}{\partial u_x^2\,\partial u}F\right)\left(\frac{\partial^3}{\partial u_x^2\,\partial x}F\right)$$

$$+2\,u_x\left(\frac{\partial^3}{\partial u_x^2\,\partial u}F\right)\left(\frac{\partial^3}{\partial u_x^3}F\right)F+\left(\frac{\partial^2}{\partial u_x\,\partial u}F\right)^2-2\left(\frac{\partial^2}{\partial u_x\,\partial u}F\right)\left(\frac{\partial^3}{\partial u_x^2\,\partial x}F\right)$$

$$-2\left(\frac{\partial^2}{\partial u_x\,\partial u}F\right)\left(\frac{\partial^3}{\partial u_x^3}F\right)F+\left(\frac{\partial^3}{\partial u_x^2\,\partial x}F\right)^2+2\left(\frac{\partial^3}{\partial u_x^2\,\partial x}F\right)\left(\frac{\partial^3}{\partial u_x^3}F\right)F$$

$$+\left(\frac{\partial^3}{\partial u_x^3}F\right)^2F^2\Bigg),\ a_4=-\frac{1}{2}\left(-\left(\frac{\partial^2}{\partial u_x\,\partial u}F\right)+\frac{\partial^3}{\partial u_x^2\,\partial x}F+\left(\frac{\partial^3}{\partial u_x^3}F\right)F\right.$$
$$\left.+u_x\left(\frac{\partial^3}{\partial u_x^2\,\partial u}F\right)\right)\Bigg/\Bigg(RootOf\Bigg(-\left(\frac{\partial^3}{\partial u_x^3}F\right)u_x\left(\frac{\partial^3}{\partial u_x^2\,\partial u}F\right)+\left(\frac{\partial^3}{\partial u_x^3}F\right)\left(\frac{\partial^2}{\partial u_x\,\partial u}F\right)$$
$$-\left(\frac{\partial^3}{\partial u_x^3}F\right)\left(\frac{\partial^3}{\partial u_x^2\,\partial x}F\right)-\left(\frac{\partial^3}{\partial u_x^3}F\right)^2F+4\,\_Z^3\Bigg)\Bigg)\Bigg\}$$

```
> exterior(op(%));
```

With this assignment, θ is an invariant coframe. The structure invariants are

```
> Inv:=InvariantStructure(theta,Phi,tableau):
> nops(%);
```

(41)

$$7 \qquad \textbf{(41)}$$

```
> map(length,Inv);
```

$$[1699, 1699, 4521, 4521, 14477, 15637, 23199] \qquad \textbf{(42)}$$

```
> Inv[1];
```

$$\frac{1}{6}\left(\left(\left(\frac{\partial^4}{\partial u_x^4}F\right)u_x\left(\frac{\partial^3}{\partial u_x^2\,\partial u}F\right)+\left(\frac{\partial^3}{\partial u_x^3}F\right)u_x\left(\frac{\partial^4}{\partial u_x^3\,\partial u}F\right)-\left(\frac{\partial^4}{\partial u_x^4}F\right)\left(\frac{\partial^2}{\partial u_x\,\partial u}F\right)+\left(\frac{\partial^4}{\partial u_x^4}F\right)\left(\frac{\partial^3}{\partial u_x^2\,\partial x}F\right)+\left(\frac{\partial^3}{\partial u_x^3}F\right)\left(\frac{\partial^4}{\partial u_x^3\,\partial x}F\right)+2\left(\frac{\partial^3}{\partial u_x^3}F\right)F\left(\frac{\partial^4}{\partial u_x^4}F\right)+\left(\frac{\partial^3}{\partial u_x^3}F\right)^2\left(\frac{\partial}{\partial u_x}F\right)\right)\Big/\left(\mathit{RootOf}\left(-\left(\frac{\partial^3}{\partial u_x^3}F\right)u_x\left(\frac{\partial^3}{\partial u_x^2\,\partial u}F\right)+\left(\frac{\partial^3}{\partial u_x^3}F\right)\left(\frac{\partial^2}{\partial u_x\,\partial u}F\right)-\left(\frac{\partial^3}{\partial u_x^3}F\right)\left(\frac{\partial^3}{\partial u_x^2\,\partial x}F\right)-\left(\frac{\partial^3}{\partial u_x^3}F\right)^2F+4\_Z^3\right)^2\left(\frac{\partial^3}{\partial u_x^3}F\right)\right)\right) \qquad \textbf{(43)}$$

```
> Inv[2];
```

$$-\frac{1}{6}\left(2\left(\frac{\partial^3}{\partial u_x^3}F\right)^2\left(\frac{\partial}{\partial u_x}F\right)+\left(\frac{\partial^3}{\partial u_x^3}F\right)F\left(\frac{\partial^4}{\partial u_x^4}F\right)+2\left(\frac{\partial^3}{\partial u_x^3}F\right)u_x\left(\frac{\partial^4}{\partial u_x^3\,\partial u}F\right)+2\left(\frac{\partial^3}{\partial u_x^3}F\right)\left(\frac{\partial^4}{\partial u_x^3\,\partial x}F\right)-\left(\frac{\partial^4}{\partial u_x^4}F\right)u_x\left(\frac{\partial^3}{\partial u_x^2\,\partial u}F\right)+\left(\frac{\partial^4}{\partial u_x^4}F\right)\left(\frac{\partial^2}{\partial u_x\,\partial u}F\right)-\left(\frac{\partial^4}{\partial u_x^4}F\right)\left(\frac{\partial^3}{\partial u_x^2\,\partial x}F\right)\right)\Big/\left(\mathit{RootOf}\left(-\left(\frac{\partial^3}{\partial u_x^3}F\right)u_x\left(\frac{\partial^3}{\partial u_x^2\,\partial u}F\right)+\left(\frac{\partial^3}{\partial u_x^3}F\right)\left(\frac{\partial^2}{\partial u_x\,\partial u}F\right)-\left(\frac{\partial^3}{\partial u_x^3}F\right)\left(\frac{\partial^3}{\partial u_x^2\,\partial x}F\right)-\left(\frac{\partial^3}{\partial u_x^3}F\right)^2F+4\_Z^3\right)^2\left(\frac{\partial^3}{\partial u_x^3}F\right)\right) \qquad \textbf{(44)}$$

```
> simplify(2*Inv[1]+Inv[2]);
```

$$\frac{1}{2}\left(\left(\frac{\partial^4}{\partial u_x^4}F\right)\left(-\left(\frac{\partial^2}{\partial u_x\,\partial u}F\right)+\frac{\partial^3}{\partial u_x^2\,\partial x}F+\left(\frac{\partial^3}{\partial u_x^3}F\right)F+u_x\left(\frac{\partial^3}{\partial u_x^2\,\partial u}F\right)\right)\right)\Big/\left(\mathit{RootOf}\left(-\left(\frac{\partial^3}{\partial u_x^3}F\right)u_x\left(\frac{\partial^3}{\partial u_x^2\,\partial u}F\right)+\left(\frac{\partial^3}{\partial u_x^3}F\right)\left(\frac{\partial^2}{\partial u_x\,\partial u}F\right)-\left(\frac{\partial^3}{\partial u_x^3}F\right)\left(\frac{\partial^3}{\partial u_x^2\,\partial x}F\right)-\left(\frac{\partial^3}{\partial u_x^3}F\right)^2F+4\_Z^3\right)^2\left(\frac{\partial^3}{\partial u_x^3}F\right)\right) \qquad \textbf{(45)}$$

Unfortunately they are not all independent (there are 5 independent invariants) and not in "optimal" form. A job for the next version! For the equation $u_{x,x} = e^{u_x}$, the invariants are

```
> eval(Inv,F=exp(u[x]));
```

$$\left[\frac{1}{2}\,\frac{\left(e^{u_x}\right)^2}{RootOf\left(-\left(e^{u_x}\right)^3+4\,\_Z^3\right)^2},\ -\frac{1}{2}\,\frac{\left(e^{u_x}\right)^2}{RootOf\left(-\left(e^{u_x}\right)^3+4\,\_Z^3\right)^2},\ \frac{1}{2}\,\frac{\left(e^{u_x}\right)^2}{RootOf\left(-\left(e^{u_x}\right)^3+4\,\_Z^3\right)^2},\ \frac{1}{2}\,\frac{\left(e^{u_x}\right)^2}{RootOf\left(-\left(e^{u_x}\right)^3+4\,\_Z^3\right)^2},\ 0,\ 0,\ 0\right] \tag{46}$$

```
> convert(%,radical):
> simplify(%,symbolic);
```

$$\left[2^{1/3},\ -2^{1/3},\ 2^{1/3},\ 2^{1/3},\ 0,\ 0,\ 0\right] \tag{47}$$

and so this equation cannot be linerized by a fibre preserving transformation.